

## Square Peg, Square Hole—Getting Tables to Fit on Slides in the ODS Destination for PowerPoint

Jane Eslinger, SAS Institute Inc.

### ABSTRACT

An output table is a square. A slide in Microsoft PowerPoint is a square. The table, being the smaller square, should fit in the bigger square slide. Right? Well, not always. Despite the programmer's expectations, some tables will not fit on the slide created by the ODS destination for PowerPoint. It depends on the table. For instance, tables with, say, more than 10 rows or more than 6 columns might end up spanning multiple slides. But, just as with the popular children's toy, by twisting, turning, or approaching the hole from a different angle, you can get the peg in the hole.

This paper discusses three programming strategies for getting your tables to fit on slides:

- Changing style attributes to decrease the amount of space needed for the table
- Strategically dividing one table into multiple tables
- Using ODS output data sets for greater control over the structure of the tables

Throughout this paper, you will see examples that demonstrate how to apply these strategies using the popular procedures TABULATE, REPORT, FREQ, and GLM.

### INTRODUCTION

The ODS destination for PowerPoint enables you to send SAS® tables and graphs directly to a Microsoft PowerPoint file. But it is unlike any of the other ODS destinations. The most immediate and noticeable difference for programmers is the physical size of the tables. PowerPoint slides are intended to be viewed on a relatively large display. Consequently, the default font size for tables created by the ODS destination for PowerPoint is 20 point. In contrast, the default font size for ODS PDF output is 8 point.

Tables that fit easily in a paging destination such as PDF do not necessarily fit on a slide created by the ODS destination for PowerPoint. Additionally, the PowerPoint destination is not able to create all of the tables that you can create manually in Microsoft PowerPoint.

This paper focuses on strategies for getting tables from the TABULATE, REPORT, FREQ, and GLM procedures to fit well and look good on a slide. Though the strategies are presented for just these four procedures, they can be applied to any other procedure you might use. The strategies include changing attributes and restructuring the table so that it requires less space; dividing large tables into multiple smaller tables; and taking advantage of ODS output data sets to judiciously select or exclude tables.

While the strategies are simple, the execution requires time and some trial and error. Be aware that getting a table to fit on just one slide might not be possible. It depends on the table. If the table will not fit on one slide, you need to focus on getting the table to split where you want it to, making it as easy to consume as possible.

### SQUARE PEG – PROC TABULATE

PROC TABULATE enables you to specify column, row, and page dimensions. The tables it creates might be both long and wide, which can make fitting them onto a slide difficult. The TABULATE output you create works well for other destinations, but it might not work well for the ODS destination for PowerPoint. The dimensions and overall size of the table depend on the following:

- the number of CLASS variables
- the number of categories within each CLASS variable

- the number of analysis variables
- the number of statistics requested for each analysis variable.

Example 1 demonstrates a typical PROC TABULATE step: two CLASS variables, two analysis variables, and three statistics.

**Example 1:**

```
ods powerpoint file='example1.pptx';
title 'Tabulate: Default';
proc tabulate data=sashelp.prdsal3;
  class country product;
  var actual predict;
  table country*product,(actual predict)*(n sum mean);
run;
ods powerpoint close;
```

The resulting table, shown in Output 1, spans two slides. It also breaks Mexico across two slides, which is not ideal.

Country	Product	Actual Sales			Predicted Sales		
		N	Sum	Mean	N	Sum	Mean
Canada	BED	576	460670.80	799.78	576	496266.70	861.57
	CHAIR	576	457739.30	794.69	576	494753.40	858.95
	DESK	576	469536.10	815.17	576	492749.30	855.47
	SOFA	576	477230.10	828.52	576	518489.40	900.16
Mexico	BED	576	312039.10	541.73	576	353177.90	613.16
	CHAIR	576	300830.80	522.28	576	333331.50	578.70
	DESK	576	304936.20	529.40	576	337728.40	586.33

(Continued)

**Output 1. PROC TABULATE Default**

**STRATEGY: REMOVE OPTIONAL TEXT**

The first strategy for getting a PROC TABULATE table to fit better on the slides is to remove the '(Continued)' text. The strategy, which is straightforward and does not alter the meaning of the table, is to add the NOCONTINUED option to the TABLE statement as shown in Example 2.

**Example 2:**

```
ods powerpoint file='example2.pptx';
title 'Tabulate: NOCONTINUED Option';
proc tabulate data=sashelp.prdsal3;
  class country product;
  var actual predict;
  table country*product,(actual predict)*(n sum mean) / nocontinued;
run;
ods powerpoint close;
```

As you can see in Output 2, the removal of the '(Continued)' text makes enough room for the Mexico section to fit on the first slide. The table still spans two slides, but it is easier to read now that the information for each country is contained on the same slide.

**Tabulate: NOCONTINUED option**

Country	Product	Actual Sales			Predicted Sales		
		N	Sum	Mean	N	Sum	Mean
Canada	BED	576	460670.80	799.78	576	496266.70	861.57
	CHAIR	576	457739.30	794.69	576	494753.40	858.95
	DESK	576	469536.10	815.17	576	492749.30	855.47
	SOFA	576	477230.10	828.52	576	518489.40	900.16
Mexico	BED	576	312039.10	541.73	576	353177.90	613.16
	CHAIR	576	300830.80	522.28	576	333331.50	578.70
	DESK	576	304936.20	529.40	576	337728.40	586.33
	SOFA	576	313853.00	544.88	576	350385.50	608.31

**Tabulate: NOCONTINUED option**

U.S.A.	Product	Actual Sales			Predicted Sales		
		N	Sum	Mean	N	Sum	Mean
U.S.A.	BED	1728	1823734.20	1055.40	1728	1894823.20	1096.54
	CHAIR	1728	1838074.70	1063.70	1728	1953700.50	1130.61
	DESK	1728	1830805.80	1059.49	1728	1969057.70	1139.50
	SOFA	1728	1927313.00	1115.34	1728	1937251.00	1121.09

**Output 2. PROC TABULATE NOCONTINUED**

**STRATEGY: REDUCE TABLE SIZE**

Another approach is to reduce the amount of space required by the table. The BOX= procedure option enables you to add text to the header section above the columns that make up the row dimension. Utilizing this option enables you to remove the labels for the row dimension variables. Example 3 eliminates the labels for the COUNTRY and PRODUCT variables and adds the appropriate text via the BOX= option.

**Example 3:**

```
ods powerpoint file='example3.pptx';
title 'Tabulate: BOX=';
proc tabulate data=sashelp.prdsal3;
  class country product;
  var actual predict;
  table country=' '*product='',(actual predict)*(n sum mean) /
    nocontinued box='Country*Product';
run;
ods powerpoint close;
```

As shown in Output 3, removing labels and taking advantage of BOX= provides space for one more row of the table to fit on the first slide. It does not drastically change the results in this example. But if you find that you need space for just one more row of data, this might be a good strategy for you to implement.

**Tabulate: BOX=**

Country*Product		Actual Sales			Predicted Sales		
		N	Sum	Mean	N	Sum	Mean
Canada	BED	576	460670.80	799.78	576	496266.70	861.57
	CHAIR	576	457739.30	794.69	576	494753.40	858.95
	DESK	576	469536.10	815.17	576	492749.30	855.47
	SOFA	576	477230.10	828.52	576	518489.40	900.16
Mexico	BED	576	312039.10	541.73	576	353177.90	613.16
	CHAIR	576	300830.80	522.28	576	333331.50	578.70
	DESK	576	304936.20	529.40	576	337728.40	586.33
	SOFA	576	313853.00	544.88	576	350385.50	608.31
U.S.A.	BED	1728	1823734.20	1055.40	1728	1894823.20	1096.54

**Tabulate: BOX=**

Country*Product		Actual Sales			Predicted Sales		
		N	Sum	Mean	N	Sum	Mean
U.S.A.	CHAIR	1728	1838074.70	1063.70	1728	1953700.50	1130.61
	DESK	1728	1830805.80	1059.49	1728	1969057.70	1139.50
	SOFA	1728	1927313.00	1115.34	1728	1937251.00	1121.09

**Output 3. PROC TABULATE BOX=**

## STRATEGY: CHANGE FONT SIZE

Within PROC TABULATE, the STYLE= option is available in multiple statements. Using this option, you can change the value of attributes, such as the FONT\_SIZE= style attribute, for a given table. Using a smaller font size creates a smaller table and therefore more data rows can fit on one slide.

The challenge with changing attributes within PROC TABULATE is knowing which statement to use to change each part of the report. Example 4 contains the STYLE= (or S=) option in multiple statements.

- In the PROC TABULATE statement, the STYLE= option affects the data in the table cells.
- In the CLASS statement, the STYLE= option affects the CLASS variable headers.
- In the CLASSLEV statement, the STYLE= option affects the CLASS variable values.
- In the VAR statement, the STYLE= option affects the variable headers.
- As an option in the TABLE statement, the STYLE= option can be associated with the BOX= option so that it affects that text.
- In the KEYWORD statement, the STYLE= option affects the statistics' headers.

### Example 4:

```
ods powerpoint file='example4.pptx';
title 'Tabulate: Change Font Size';
proc tabulate data=sashelp.prdsal3 style=[font_size=16pt];
  class country product / style=[font_size=16pt];
  classlev country product / style=[font_size=16pt];
  var actual predict / style=[font_size=16pt];
  table country=' '*product='',(actual predict)*(n sum mean) /
    nocontinued box=[label='Country*Product' s=[font_size=16pt]];
  keyword n sum mean / style=[font_size=16pt];
run;
ods powerpoint close;
```

Output 4 shows how the smaller font size changes the overall size of the table.

Tabulate: Change Font Size							
Country*Product		Actual Sales			Predicted Sales		
		N	Sum	Mean	N	Sum	Mean
Canada	BED	576	460670.80	799.78	576	496266.70	861.57
	CHAIR	576	457739.30	794.69	576	494753.40	858.95
	DESK	576	469536.10	815.17	576	492749.30	855.47
	SOFA	576	477230.10	828.52	576	518489.40	900.16
Mexico	BED	576	312039.10	541.73	576	353177.90	613.16
	CHAIR	576	300830.80	522.28	576	333331.50	578.70
	DESK	576	304936.20	529.40	576	337728.40	586.33
	SOFA	576	313853.00	544.88	576	350385.50	608.31
U.S.A.	BED	1728	1823734.20	1055.40	1728	1894823.20	1096.54
	CHAIR	1728	1838074.70	1063.70	1728	1953700.50	1130.61

Tabulate: Change Font Size							
Country*Product		Actual Sales			Predicted Sales		
		N	Sum	Mean	N	Sum	Mean
U.S.A.	DESK	1728	1830805.80	1059.49	1728	1969057.70	1139.50
	SOFA	1728	1927313.00	1115.34	1728	1937251.00	1121.09

### Output 4. PROC TABULATE Change Font Size

You might be wondering why the table still spans two slides. It looks like the two rows on the second slide could easily fit on the first slide. In order to preserve the buffer for the footnote and prevent the table from colliding with the footnote box, the destination breaks the table and places part of it on a subsequent slide.

**Note:** For presentation purposes, the author cautions against reducing the font size below 16 point. The data on the slides should be readable by an audience viewing the presentation on a large display.

## STRATEGY: CREATE MULTIPLE SMALLER TABLES

So far, our strategies, while useful, get us only a little extra space. But you might need more than just a little extra space. If your table is long, or you do not like where it breaks on the slides, try this final strategy: include a page dimension to break large tables into smaller tables, placing each table on its own slide.

Example 5 moves COUNTRY from the row dimension to the page dimension. This move is accomplished by adding a comma between the COUNTRY and PRODUCT variables in the TABLE statement.

### Example 5:

```
ods powerpoint file='example5.pptx';
title 'Tabulate: Page Dimension';
proc tabulate data=sashelp.prdsal3;
  class country product;
  var actual predict;
  table country,product,(actual predict)*(n sum mean);
run;
ods powerpoint close;
```

The result, shown in Output 5, is one slide per country.

Country Canada

Product	Actual Sales			Predicted Sales		
	N	Sum	Mean	N	Sum	Mean
BED	576	460670.80	799.78	576	496266.70	861.57
CHAIR	576	457739.30	794.69	576	494753.40	858.95
DESK	576	469536.10	815.17	576	492749.30	855.47
SOFA	576	477230.10	828.52	576	518489.40	900.16

1

### Output 5. PROC TABULATE Page Dimension (Slide 1 of 3)

**Note:** A BY statement produces a result similar to the page dimension because it, too, generates a new table for each BY group.

These strategies for fitting your PROC TABULATE peg into the square slide hole are easy to test, do not require a major rewrite of your code, and can make a dramatic difference on the resulting slides.

## SQUARE PEG – PROC REPORT

PROC REPORT is often the go-to procedure for creating final reports because it gives you a good degree of control over the styling aspects of the report. However, the results of PROC REPORT might not fit the way you want them to on a slide created by the ODS destination for PowerPoint.

Example 6 demonstrates a typical PROC REPORT step: three GROUP variables, an ACROSS variable, and two analysis variables.

### Example 6:

```
ods powerpoint file='example6.pptx';
title 'Report: Default';
proc report data=sashelp.prdsal3;
  column year country product quarter,(actual predict);
  define year / group;
  define country / group;
  define product / group;
  define quarter / across order=internal;
  define actual / format=dollar12.2;
  define predict / format=dollar12.2;
  break after country / summarize;
run;
ods powerpoint close;
```

Output 6 shows two of the 12 total slides needed to display the entire table. As you can see, the length and width of the table prevents the entire table from fitting on one slide (or even two). However, it can be improved.

			Quarter	
			1	
Year	Country	Product	Actual Sales	Predicted Sales
1997	Canada	BED	\$50,364.00	\$59,360.00
		CHAIR	\$54,012.00	\$64,092.00
		DESK	\$58,515.00	\$61,591.00
		SOFA	\$63,835.00	\$60,529.00
1997	Canada		\$226,726.00	\$245,572.00
	Mexico	BED	\$31,422.00	\$43,409.00
		CHAIR	\$32,797.00	\$41,075.00
		DESK	\$34,808.00	\$38,706.00

				Quarter	
		2		3	
Actual Sales	Predicted Sales	Actual Sales	Predicted Sales	Actual Sales	Predicted Sales
\$56,003.00	\$55,824.00	\$56,146.00	\$63,698.00		
\$48,908.00	\$49,603.00	\$46,842.00	\$59,349.00		
\$54,593.00	\$59,480.00	\$55,635.00	\$57,553.00		
\$56,004.00	\$64,311.00	\$56,666.00	\$58,788.00		
\$215,508.00	\$229,218.00	\$215,289.00	\$239,388.00		
\$39,270.00	\$42,041.00	\$41,493.00	\$43,902.00		
\$36,412.00	\$38,515.00	\$35,277.00	\$38,581.00		
\$38,632.00	\$38,150.00	\$36,269.00	\$41,340.00		

Output 6. PROC REPORT Default (Slides 1 and 2 of 12)

### STRATEGY: REDUCE THE NUMBER OF HEADER SECTION ROWS

The header section is a great place to start when modifying a PROC REPORT table. You can reduce the three default rows to two by creating a format to apply to the QUARTER variable. Creating a format that contains 'Q', 'QTR', or 'QUARTER' enables you to remove the label, which eliminates one of the header rows.

Example 7 builds a format for QUARTER.

### Example 7:

```
ods powerpoint file='example7.pptx';
title 'Report: Reduce Header Rows';

proc format;
  value myqtr 1='Q1' 2='Q2' 3='Q3' 4='Q4';

proc report data=sashelp.prdsal3;
  column year country product quarter,(actual predict);
  define year / group;
  define country / group;
  define product / group;
  define quarter / across order=internal format=myqtr. '';
  define actual / format=dollar12.2;
  define predict / format=dollar12.2;
  break after country / summarize;
run;
ods powerpoint close;
```

The result, shown in Output 7, is a table that takes one fewer slide.

		Q1		
Year	Country	Product	Actual Sales	Predicted Sales
1997	Canada	BED	\$50,364.00	\$59,360.00
		CHAIR	\$54,012.00	\$64,092.00
		DESK	\$58,515.00	\$61,591.00
		SOFA	\$63,835.00	\$60,529.00
1997	Canada		\$226,726.00	\$245,572.00
	Mexico	BED	\$31,422.00	\$43,409.00
		CHAIR	\$32,797.00	\$41,075.00
		DESK	\$34,808.00	\$38,706.00
		SOFA	\$36,228.00	\$37,079.00

		Q2		Q3		
Year	Country	Product	Actual Sales	Predicted Sales	Actual Sales	Predicted Sales
			\$56,003.00	\$55,824.00	\$56,146.00	\$63,698.00
			\$48,908.00	\$49,603.00	\$46,842.00	\$59,349.00
			\$54,593.00	\$59,480.00	\$55,635.00	\$57,553.00
			\$56,004.00	\$64,311.00	\$56,666.00	\$58,788.00
			\$215,508.00	\$229,218.00	\$215,289.00	\$239,388.00
			\$39,270.00	\$42,041.00	\$41,493.00	\$43,902.00
			\$36,412.00	\$38,515.00	\$35,277.00	\$38,581.00
			\$38,632.00	\$38,150.00	\$36,269.00	\$41,340.00
			\$31,851.00	\$42,933.00	\$37,060.00	\$40,061.00

Output 7. PROC REPORT Reduces the Number of Header Rows (Slides 1 and 2 of 11)

### STRATEGY: USE SPANNING HEADERS

For PROC REPORT, you should also consider moving column headers into the spanning header role. Depending on the table, the headers, and your business needs, this change might enable you to remove the row with the individual column headers altogether.

For spanning headers to be effective, individual column headers must be set to blanks. When all column headers are set to blank, ODS destinations remove that header row, thus reducing the size of the header section. This strategy can be a little tricky because you need to make sure that the audience still understands what numbers are present in the table.

Example 8 modifies the format labels to contain text regarding the ACTUAL and PREDICT columns. This example places spanning headers in the COLUMN statement for the YEAR, COUNTRY, and PRODUCT variables and assigns a blank column header in each DEFINE statement.

### Example 8:

```
ods powerpoint file='example8.pptx';
title 'Report: Use Spanning Headers';

proc format;
  value myqtr 1='Q1: Actual | Predicted'
             2='Q2: Actual | Predicted'
             3='Q3: Actual | Predicted'
             4='Q4: Actual | Predicted';

proc report data=sashelp.prdsal3;
  column ('Year' year) ('Country' country) ('Product' product)
         quarter,(actual predict);
  define year / group '';
  define country / group '';
  define product / group '';
  define quarter / across order=internal format=myqtr. '';
  define actual / format=dollar12.2 '';
  define predict / format=dollar12.2 '';
  break after country / summarize;
run;
ods powerpoint close;
```

The table in Output 8 now contains just one header row.

Year	Country	Product	Q1: Actual	Predicted
1997	Canada	BED	\$50,364.00	\$59,360.00
		CHAIR	\$54,012.00	\$64,092.00
		DESK	\$58,515.00	\$61,591.00
		SOFA	\$63,835.00	\$60,529.00
1997	Canada		\$226,726.00	\$245,572.00
	Mexico	BED	\$31,422.00	\$43,409.00
		CHAIR	\$32,797.00	\$41,075.00
		DESK	\$34,808.00	\$38,706.00
		SOFA	\$36,228.00	\$37,079.00
1997	Mexico		\$135,255.00	\$160,269.00

### Output 8. PROC REPORT Spanning Headers (Slide 1 of 9)

#### STRATEGY: DISPLAY THE VALUE AS A ROW INSTEAD OF A COLUMN

To make the table both fit on the slide and look better, another strategy is to remove the column that contains a GROUP or ORDER variable and display its value as a row instead. Example 9 uses the NOPRINT option to suppress the YEAR column from appearing. Its value is written out via a LINE

statement. Notice that the example includes the ID option in the DEFINE statement for PRODUCT. The ID option ensures that the COUNTRY and PRODUCT columns print on the second slide when the table breaks across slides.

**Example 9:**

```
ods powerpoint file='example9.pptx';
title 'Report: Use LINE statement';

proc format;
  value myqtr 1='Q1: Actual | Predicted'
             2='Q2: Actual | Predicted'
             3='Q3: Actual | Predicted'
             4='Q4: Actual | Predicted';

proc report data=sashelp.prdsal3;
  column year country product quarter,(actual predict);
  define year / group noprint;
  define country / group '';
  define product / group '' id;
  define quarter / across order=internal format=myqtr. '';
  define actual / format=dollar12.2 '';
  define predict / format=dollar12.2 '';
  break after country / summarize;

  compute before year;
    line year 4.;
  endcomp;
run;
ods powerpoint close;
```

The resulting table, in Output 9, is now only seven slides rather than the original 12 in Output 6. Another improvement is that each slide now holds two quarters' worth of data. Remember that some of these strategies are about making the table easier to read and consume, more so than what is generated by default.

Report: Use LINE statement					
Country	Product	Q1: Actual	Predicted	Q2: Actual	Predicted
1997					
Canada	BED	\$50,364.00	\$59,360.00	\$56,003.00	\$55,824.00
	CHAIR	\$54,012.00	\$64,092.00	\$48,908.00	\$49,603.00
	DESK	\$58,515.00	\$61,591.00	\$54,593.00	\$59,480.00
	SOFA	\$63,835.00	\$60,529.00	\$56,004.00	\$64,311.00
<i>Canada</i>		<i>\$226,726.00</i>	<i>\$245,572.00</i>	<i>\$215,508.00</i>	<i>\$229,218.00</i>
Mexico	BED	\$31,422.00	\$43,409.00	\$39,270.00	\$42,041.00
	CHAIR	\$32,797.00	\$41,075.00	\$36,412.00	\$38,515.00
	DESK	\$34,808.00	\$38,706.00	\$38,632.00	\$38,150.00
	SOFA	\$36,228.00	\$37,079.00	\$31,851.00	\$42,933.00

  

Report: Use LINE statement					
Country	Product	Q3: Actual	Predicted	Q4: Actual	Predicted
1997					
Canada	BED	\$56,146.00	\$63,698.00	\$61,649.00	\$56,567.00
	CHAIR	\$46,842.00	\$59,349.00	\$52,521.00	\$57,947.00
	DESK	\$55,635.00	\$57,553.00	\$61,927.00	\$56,821.00
	SOFA	\$56,666.00	\$58,788.00	\$56,931.00	\$59,219.00
<i>Canada</i>		<i>\$215,289.00</i>	<i>\$239,388.00</i>	<i>\$233,028.00</i>	<i>\$230,554.00</i>
Mexico	BED	\$41,493.00	\$43,902.00	\$39,363.00	\$40,380.00
	CHAIR	\$35,277.00	\$38,581.00	\$36,814.00	\$40,420.00
	DESK	\$36,269.00	\$41,340.00	\$37,496.00	\$39,381.00
	SOFA	\$37,060.00	\$40,061.00	\$37,202.00	\$41,657.00

**Output 9. PROC REPORT LINE Statement (Slides 1 and 2 of 7)**

This LINE statement strategy is most effective when vertical space is not as much of a concern or you are going to use the PAGE option to divide the table between logical groupings.

## STRATEGY: CHANGE FONT SIZE

The next strategy is to change the font size of the header and data rows of the table. Remember, the default font size for tables is 20 point. You can change the font size to reduce the size of the table, yet keep it readable by your audience.

Example 10 changes the font size from 20 point to 16 point for the headers and data. The font size for the LINE statement is not changed in this example. However, you can change it via a STYLE= option in the COMPUTE statement.

### Example 10:

```
ods powerpoint file='example10.pptx';
title 'Report: Change Font Size';

proc format;
  value myqtr 1='Q1' 2='Q2' 3='Q3' 4='Q4';

proc report data=sashelp.prdsal3 style(header column)=[fontsize=16pt];
  column year country product quarter,(actual predict);
  define year / group noprint;
  define country / group;
  define product / group id;
  define quarter / across order=internal format=myqtr. '';
  define actual / format=dollar12.2;
  define predict / format=dollar12.2;
  break after country / summarize;

  compute before year;
    line year 4.;
  endcomp;
run;
ods powerpoint close;
```

Changing the font size reduced the number of slides required to display the table. As indicated in Output 10, the table now needs only six slides.

Report: Change Font Size					
		Q1		Q2	
Country	Product	Actual Sales	Predicted Sales	Actual Sales	Predicted Sales
1997					
Canada	BED	\$50,364.00	\$59,360.00	\$56,003.00	\$55,824.00
	CHAIR	\$54,012.00	\$64,092.00	\$48,908.00	\$49,603.00
	DESK	\$58,515.00	\$61,591.00	\$54,593.00	\$59,480.00
	SOFA	\$63,835.00	\$60,529.00	\$56,004.00	\$64,311.00
Canada		\$226,726.00	\$245,572.00	\$215,508.00	\$229,218.00
Mexico	BED	\$31,422.00	\$43,409.00	\$39,270.00	\$42,041.00
	CHAIR	\$32,797.00	\$41,075.00	\$36,412.00	\$38,515.00
	DESK	\$34,808.00	\$38,706.00	\$38,632.00	\$38,150.00
	SOFA	\$36,228.00	\$37,079.00	\$31,851.00	\$42,933.00
Mexico		\$135,255.00	\$160,269.00	\$146,165.00	\$161,639.00

Report: Change Font Size					
		Q3		Q4	
Country	Product	Actual Sales	Predicted Sales	Actual Sales	Predicted Sales
1997					
Canada	BED	\$56,146.00	\$63,698.00	\$61,649.00	\$56,567.00
	CHAIR	\$46,842.00	\$59,349.00	\$52,521.00	\$57,947.00
	DESK	\$55,635.00	\$57,553.00	\$61,927.00	\$56,821.00
	SOFA	\$56,666.00	\$58,788.00	\$56,931.00	\$59,219.00
Canada		\$215,289.00	\$239,388.00	\$233,028.00	\$230,554.00
Mexico	BED	\$41,493.00	\$43,902.00	\$39,363.00	\$40,380.00
	CHAIR	\$35,277.00	\$38,581.00	\$36,814.00	\$40,420.00
	DESK	\$36,269.00	\$41,340.00	\$37,496.00	\$39,381.00
	SOFA	\$37,060.00	\$40,061.00	\$37,202.00	\$41,657.00
Mexico		\$150,099.00	\$163,884.00	\$150,875.00	\$161,838.00

### Output 10. PROC REPORT Font Size (Slides 1 and 2 of 6)

**Note:** For presentation purposes, the author cautions against reducing the font size below 16 point. The data on the slides should be readable by an audience viewing the presentation on a large display.

## STRATEGY: CREATE MULTIPLE SMALLER TABLES

The final strategy for PROC REPORT is the same strategy shown earlier for PROC TABULATE—to create a page dimension, which in PROC REPORT is often referred to as a paging variable. In the PRDSAL3 data set, you can use the YEAR variable as a paging variable.

Example 11 includes a BREAK statement with the PAGE option. The compute block now uses the `_PAGE_` target so that the value of YEAR is placed above the table.

### Example 11:

```
ods powerpoint file='example11.pptx';
title 'Report: Paging Variable';
proc report data=sashelp.prdsal3 style(header column)=[fontsize=16pt];
  column year country product quarter,(actual predict);
  define year / group noprint;
  define country / group;
  define product / group id;
  define quarter / across order=internal format=myqtr. '';
  define actual / format=dollar12.2;
  define predict / format=dollar12.2;
  break after country / summarize;
  break after year / page;

  compute before _page_;
    line year 4.;
  endcomp;
run;
ods powerpoint close;
```

The table generated by the Example 11 code, and shown in Output 11, requires more slides. However, the goal of using a paging variable is to make the table easier to understand, not to reduce the number of slides.

1997					
		Q1		Q2	
Country	Product	Actual Sales	Predicted Sales	Actual Sales	Predicted Sales
Canada	BED	\$50,364.00	\$59,360.00	\$56,003.00	\$55,824.00
	CHAIR	\$54,012.00	\$64,092.00	\$48,908.00	\$49,603.00
	DESK	\$58,515.00	\$61,591.00	\$54,593.00	\$59,480.00
	SOFA	\$63,835.00	\$60,529.00	\$56,004.00	\$64,311.00
<i>Canada</i>		<i>\$226,726.00</i>	<i>\$245,572.00</i>	<i>\$215,508.00</i>	<i>\$229,218.00</i>
Mexico	BED	\$31,422.00	\$43,409.00	\$39,270.00	\$42,041.00
	CHAIR	\$32,797.00	\$41,075.00	\$36,412.00	\$38,515.00
	DESK	\$34,808.00	\$38,706.00	\$38,632.00	\$38,150.00
	SOFA	\$36,228.00	\$37,079.00	\$31,851.00	\$42,933.00
<i>Mexico</i>		<i>\$135,255.00</i>	<i>\$160,269.00</i>	<i>\$146,165.00</i>	<i>\$161,639.00</i>

1

Output 11. PROC REPORT Paging Variable (Slide 1 of 8)

## SQUARE PEG – PROC FREQ

PROC FREQ generates counts and percentages as well as runs statistical tests and generates plots. This section concentrates on the tables that PROC FREQ generates with counts. For tips about how to handle output that includes statistical tests, refer to the next section on PROC GLM.

Example 12 creates a simple cross-tabulation table. COUNTRY has three unique categories, and PRODUCT has four unique categories.

### Example 12:

```
ods powerpoint file='example12.pptx';
title 'Freq: Default';
proc freq data=sashelp.prdsal3;
  tables country*product / list;
run;
ods powerpoint close;
```

The resulting table spans two slides, shown in Output 12. The output also contains the procedure title, something we did not see with the TABULATE or REPORT procedures.

Slide 1: Freq: Default

The FREQ Procedure

COUNTRY	PRODUCT	Frequency	Percent	Cumulative	
				Frequency	Percent
Canada	BED	576	5.00	576	5.00
Canada	CHAIR	576	5.00	1152	10.00
Canada	DESK	576	5.00	1728	15.00
Canada	SOFA	576	5.00	2304	20.00
Mexico	BED	576	5.00	2880	25.00
Mexico	CHAIR	576	5.00	3456	30.00
Mexico	DESK	576	5.00	4032	35.00
Mexico	SOFA	576	5.00	4608	40.00
U.S.A.	BED	1728	15.00	6336	55.00

Slide 2: Freq: Default

The FREQ Procedure

COUNTRY	PRODUCT	Frequency	Percent	Cumulative	
				Frequency	Percent
U.S.A.	CHAIR	1728	15.00	8064	70.00
U.S.A.	DESK	1728	15.00	9792	85.00
U.S.A.	SOFA	1728	15.00	11520	100.00

### Output 12. PROC FREQ Default

### STRATEGY: REMOVE OPTIONAL TEXT

The first strategy, and an easy win for making additional room for the PROC FREQ table, is to eliminate the procedure title by adding the ODS NOPROCTITLE statement.

### Example 13:

```
ods powerpoint file='example13.pptx';
title 'Freq: ODS NOPROCTITLE';
ods noproctitle;
proc freq data=sashelp.prdsal3;
  tables country*product / list;
run;
ods powerpoint close;
```

This simple strategy made room for another row of data on the first slide, shown in Output 13.

**Freq: ODS NOPROCTITLE**

COUNTRY	PRODUCT	Frequency	Percent	Cumulative Frequency	Cumulative Percent
Canada	BED	576	5.00	576	5.00
Canada	CHAIR	576	5.00	1152	10.00
Canada	DESK	576	5.00	1728	15.00
Canada	SOFA	576	5.00	2304	20.00
Mexico	BED	576	5.00	2880	25.00
Mexico	CHAIR	576	5.00	3456	30.00
Mexico	DESK	576	5.00	4032	35.00
Mexico	SOFA	576	5.00	4608	40.00
U.S.A.	BED	1728	15.00	6336	55.00
U.S.A.	CHAIR	1728	15.00	8064	70.00

**Freq: ODS NOPROCTITLE**

COUNTRY	PRODUCT	Frequency	Percent	Cumulative Frequency	Cumulative Percent
U.S.A.	DESK	1728	15.00	9792	85.00
U.S.A.	SOFA	1728	15.00	11520	100.00

### Output 13. PROC FREQ NOPROCTITLE

#### STRATEGY: UTILIZE OUTPUT DATA SETS

Your table might be longer or more complicated. In this case, you might need to try the strategies from the TABULATE and REPORT sections. PROC FREQ does not have the granular control that the other procedures do. For example, PROC FREQ does not have the ability to change style attributes within the procedure.

The strategy for handling PROC FREQ tables that do not fit well on slides is to send the data to output data sets. Then, use another procedure, like PRINT or REPORT, to place the data in the presentation.

Example 14 includes the following features:

- The ODS EXCLUDE statement prevents the PROC FREQ output from being sent to the open ODS destinations.
- The ODS OUTPUT statement captures the results of PROC FREQ in a data set.
- The LABEL option in the PROC PRINT statement ensures that the variables' labels are used.
- The STYLE= option in the PROC PRINT statement changes the font size for the entire table to 16 point.

#### Example 14:

```
ods powerpoint file='example14.pptx';
ods exclude all;
ods output list=freqout;
proc freq data=sashelp.prdsal3;
    tables country*product / list;
run;
ods select all;

title 'Freq: ODS OUTPUT';
proc print data=freqout noobs label style(header data)=[font_size=16pt];
    var country product frequency percent cumfrequency cumpercent;
    label cumfrequency='Cumulative Frequency'
          cumpercent='Cumulative Percent';
run;
ods powerpoint close;
```

The results from PROC FREQ, shown in Output 14, fit nicely on one slide by using the strategy of first creating an ODS OUTPUT data set. As mentioned previously, implementing these strategies takes some trial and error, just like a kid trying to fit a peg in a hole. In fact, the table from PROC PRINT does not fit

on one slide when you set the font size to 18 point. As with the REPORT and TABULATE output, sometimes the goal should be to make the table consumable.

**Freq: ODS OUTPUT**

Country	Product	Frequency	Percent	Cumulative Frequency	Cumulative Percent
Canada	BED	576	5.00	576	5.00
Canada	CHAIR	576	5.00	1152	10.00
Canada	DESK	576	5.00	1728	15.00
Canada	SOFA	576	5.00	2304	20.00
Mexico	BED	576	5.00	2880	25.00
Mexico	CHAIR	576	5.00	3456	30.00
Mexico	DESK	576	5.00	4032	35.00
Mexico	SOFA	576	5.00	4608	40.00
U.S.A.	BED	1728	15.00	6336	55.00
U.S.A.	CHAIR	1728	15.00	8064	70.00
U.S.A.	DESK	1728	15.00	9792	85.00
U.S.A.	SOFA	1728	15.00	11520	100.00

1

#### Output 14. PROC FREQ ODS Output Tables

### SQUARE PEG – PROC GLM

SAS/STAT® procedures can generate a large number of output objects, depending on the options that you specify. When sent to the ODS destination for PowerPoint, the output is likely to be unwieldy. PROC GLM is one such procedure. Example 15 uses GLM, specifying two CLASS variables, both of which are specified in the MODEL statement along with the interaction term. The SOLUTION option requests the normal equations (parameter estimates).

#### Example 15:

```
ods powerpoint file='example15.pptx';
title 'GLM: Default';
proc glm data=sashelp.prdsal3;
  class state product;
  model actual= state product state*product / solution;
quit;
ods powerpoint close;
```

As you can see in Output 15, this seemingly simple, straightforward request generates 21 slides of output in the ODS destination for PowerPoint!!! Some of the tables are relatively small, with only three or four rows; however, by default, the ODS destination for PowerPoint puts each output object (table) on its own slide.

GLM: Default		
Class Level Information		
Class	Levels	Values
STATE	16	Baja California Norte British Columbia California Campeche Colorado Florida Illinois Michoacan New York North Carolina Nuevo Leon Ontario Quebec Saskatchewan Texas Washington
PRODUCT	4	BED CHAIR DESK SOFA

GLM: Default	
Number of Observations Read	11520
Number of Observations Used	11520

**Output 15. PROC GLM Default (Slides 1 and 2 of 21)**

### STRATEGY: INCLUDE ONLY NECESSARY TABLES

By utilizing an ODS TRACE ON statement and looking at the results in the log, you can determine how many different pieces of output are generated by your specific code. The previous PROC GLM step created seven output tables: ClassLevels, Nobs, OverallANOVA, FitStatistics, ModelANOVA (Type I), ModelANOVA (Type III), and ParameterEstimates. Statisticians tend to look at a subset of these for their analyses. When you are using the PowerPoint destination, it is imperative to select and output only the relevant tables.

Example 16 uses the ODS SELECT statement to choose only the most relevant tables from PROC GLM. The ModelANOVA reference includes the Type I and Type III tables.

#### Example 16:

```
ods powerpoint file='example15.pptx';
title 'GLM: ODS Select';
ods select ClassLevels Nobs OverallANOVA FitStatistics ModelANOVA;
proc glm data=sashelp.prdsal3;
  class state product;
  model actual= state product state*product / solution;
quit;
ods powerpoint close;
```

The selected tables require only six slides in PowerPoint in Output 16. The slides in Output 16 match the first six slides in Output 15. Reducing the output from 21 slides to six is quite an improvement.

**GLM: ODS Select**

Class Level Information		
Class	Levels	Values
STATE	16	Baja California Norte British Columbia California Campeche Colorado Florida Illinois Michoacan New York North Carolina Nuevo Leon Ontario Quebec Saskatchewan Texas Washington
PRODUCT	4	BED CHAIR DESK SOFA

**GLM: ODS Select**

Number of Observations Read	11520
Number of Observations Used	11520

**Output 16. PROC GLM Select (Slides 1 and 2 of 6)**

Note that there are options in the MODEL statement for controlling the output that is displayed. SS1 and SS3 are such options.

**STRATEGY: PLACE TWO TABLES ON THE SAME SLIDE**

Example 17 adds the STARTPAGE= option, new in SAS® 9.4 TS1M4, to the ODS POWERPOINT statement. Strategic use of this option enables you to reduce the number of slides needed for procedures, like GLM, that produce a lot of individual tables.

**Example 17:**

```
ods powerpoint file='example17.pptx' startpage=no;
title 'GLM: STARTPAGE=';
ods select ClassLevels Nobs OverallANOVA FitStatistics ModelANOVA;
proc glm data=sashelp.prdsal3;
  class state product;
  model actual= state product state*product / solution;
quit;
ods powerpoint close;
```

The GLM output now spans only three slides. By using these two strategies, the output was reduced from 21 slides to three. Also, the audience does not have to wade through tables that are not as relevant to the analysis.

**GLM: STARTPAGE=**

Class Level Information		
Class	Levels	Values
STATE	16	Baja California Norte British Columbia California Campeche Colorado Florida Illinois Michoacan New York North Carolina Nuevo Leon Ontario Quebec Saskatchewan Texas Washington
PRODUCT	4	BED CHAIR DESK SOFA

Number of Observations Read	11520
Number of Observations Used	11520

**GLM: STARTPAGE=**

Dependent Variable: ACTUAL Actual Sales

Source	DF	Sum of Squares	Mean Square	F Value	Pr > F
Model	63	547404377	8688958	29.18	<.0001
Error	11456	3410898437	297739		
Corrected Total	11519	3958302815			

ACTUAL				
R-Square	Coeff Var	Root MSE	Mean	
0.138293	59.77069	545.6547	912.9135	

Source	DF	Type I SS	Mean Square	F Value	Pr > F
STATE	15	535257229.3	35683815.3	119.85	<.0001
PRODUCT	3	3701733.1	1233911.0	4.14	0.0061

**Output 17. PROC GLM STARTPAGE= (Slides 1 and 2 of 3)**

It is also possible to remove the procedure note, “Dependent Variable: ACTUAL Actual Sales,” that is generated by the OverallANOVA table. However, to remove this note, you must use the TEMPLATE procedure, which is beyond the scope of this paper.

Furthermore, you can use the strategies mentioned in earlier sections of removing the procedure title or using ODS OUTPUT tables along with the PRINT or REPORT procedure to further maximize slide space.

## CONCLUSION

When a kid plays with a set of blocks, it sometimes takes more than one attempt to get a square peg to fit in a square hole. Working with tables and the ODS destination for PowerPoint is no different. The default behavior of the destination might not split the table where you want it to or it might require more slides than you think are appropriate.

By using the strategies discussed in this paper, you can make the tables in your presentation more consumable and therefore more effective.

Remember these strategies! They include the following:

- Reducing the size of the table
- Dividing large tables into multiple smaller tables
- Taking advantage of ODS OUTPUT data sets
- Selecting or excluding tables with the ODS SELECT or EXCLUDE statements

## REFERENCES

Eslinger, Jane. 2018. The SAS® Programmer's PROC REPORT Handbook: ODS Companion. Cary, NC: SAS Institute Inc.

Hunter, Tim. 2016. “A Second Look at the ODS Destination for PowerPoint.” Proceedings of the SAS Global 2016 Conference. Cary, NC: SAS Institute Inc. Available at [support.sas.com/resources/papers/proceedings16/SAS3801-2016.pdf](http://support.sas.com/resources/papers/proceedings16/SAS3801-2016.pdf).

Eslinger, Jane. 2016. “The Dynamic Duo: ODS Layout and the ODS Destination for PowerPoint.” Proceedings of the SAS Global 2016 Conference. Cary, NC: SAS Institute Inc. Available at [support.sas.com/resources/papers/proceedings16/SAS5443-2016.pdf](http://support.sas.com/resources/papers/proceedings16/SAS5443-2016.pdf).

Hunter, Tim. 2013. “A First Look at the ODS Destination for PowerPoint.” Proceedings of the SAS Global 2013 Conference. Cary, NC: SAS Institute Inc. Available at [support.sas.com/resources/papers/proceedings13/041-2013.pdf](http://support.sas.com/resources/papers/proceedings13/041-2013.pdf).

## ACKNOWLEDGMENTS

The author is immensely grateful to Bari Lawhorn and Kathryn McLawhorn, both of whom contributed to this paper. I would also like to thank Vivian McGee and Rayna Rowell for editing the paper.

## RECOMMENDED READING

- SAS® *Procedures Guide*
- SAS® 9.4 *Output Delivery System: User's Guide*
- SAS/STAT® *User's Guide*

## CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Jane Eslinger  
SAS Institute Inc.  
Cary, NC  
Email: [support@sas.com](mailto:support@sas.com)  
Web: [support.sas.com](http://support.sas.com)

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.